Recoil Polarization Bases

James J. Kelly

Department of Physics
University of Maryland
College Park, MD 20742
jjkelly@physics.umd.edu

Several bases useful for analysis of recoil polarization in $\left(\vec{e}, e' \vec{N}\right)$ reactions and their interrelationships are developed.

Last revised: November 15, 1999

Initialization

■ Defaults and packages

```
ClearAll["Global`*"];
Off[General::spell, General::spell1]

Needs["Utilities`Notation`"]

Symbolize[x̂]; Symbolize[ŷ]; Symbolize[ẑ];
Symbolize[t̂]; Symbolize[n̂]; Symbolize[ℓ̂];
Symbolize[â]; Symbolize[b̂]; Symbolize[ĉ];
Symbolize[d̂]; Symbolize[ê]; Symbolize[f̂];
Symbolize[θs];
```

■ Rotation matrices

We begin with a primitive basis in which $\hat{z} = \hat{q}$ is along the virtual photon, $\hat{y} \propto \hat{k}_i \times \hat{k}_f$ is normal to the electron scattering plane, and $\hat{x} = \hat{y} \times \hat{z}$ is transverse within that plane. It is convenient to call this the *photon* basis. Other bases can be generated by a sequence of rotations based upon the matrices

```
rotz[θ_] := {{Cos[θ], -Sin[θ], 0}, {Sin[θ], Cos[θ], 0}, {0, 0, 1}};
rotx[θ_] := {{1, 0, 0}, {0, Cos[θ], -Sin[θ]}, {0, Sin[θ], Cos[θ]}};
roty[θ_] := {{Cos[θ], 0, Sin[θ]}, {0, 1, 0}, {-Sin[θ], 0, Cos[θ]}}
```

which generate counterclockwise rotations of a vector about the specified coordinate axis. Note that these matrices produce *active* rotations of a vector wrt the original coordinate system.

Next, we define a general rotation in terms of three Euler angles representing rotation about $\hat{z}$ through angle α, followed by rotation about $\hat{y}'$ through angle β, and finally rotation about $\hat{z}''$ through angle γ. One can show that the same transformation is achieved by rotation γ about the $\hat{z}$, followed by rotation β about the original $\hat{y}$ axis, and finally rotation α about the original $\hat{z}$ axis, whereby

```
euler[α_, β_, γ_] := rotz[α].roty[β].rotz[γ]
```

```
euler[α, β, γ] // Simplify
```

```
{{Cos[α] Cos[β] Cos[γ] - Sin[α] Sin[γ],
  -Cos[γ] Sin[α] - Cos[α] Cos[β] Sin[γ], Cos[α] Sin[β]},
 {Cos[β] Cos[γ] Sin[α] + Cos[α] Sin[γ],
  Cos[α] Cos[γ] - Cos[β] Sin[α] Sin[γ], Sin[α] Sin[β]},
 {-Cos[γ] Sin[β], Sin[β] Sin[γ], Cos[β]}}
```

Several special cases are shown below, where the columns provide the transformed basis vectors in terms of the original coordinate system.

```
euler[ϕ, 0, 0] // MatrixForm
```

$$\begin{pmatrix} \text{Cos}[\phi] & -\text{Sin}[\phi] & 0 \\ \text{Sin}[\phi] & \text{Cos}[\phi] & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
euler[0, θ, 0] // MatrixForm
```

$$\begin{pmatrix} \text{Cos}[\theta] & 0 & \text{Sin}[\theta] \\ 0 & 1 & 0 \\ -\text{Sin}[\theta] & 0 & \text{Cos}[\theta] \end{pmatrix}$$

```
euler[ϕ, 0, 0].{x, y, 0}
```

```
{x Cos[ϕ] - y Sin[ϕ], y Cos[ϕ] + x Sin[ϕ], 0}
```

```
euler[0, θ, 0].{x, 0, z}
```

```
{x Cos[θ] + z Sin[θ], 0, z Cos[θ] - x Sin[θ]}
```

■ Normalized cross product

It will be useful to define a vector cross product with unit normalization.

```
NormalizedCrossProduct[a_, b_] :=
 Module[{c}, c = Cross[a, b]; Simplify[c / √c.c]]
```

■ Transformation matrices

The following function constructs the transformation matrix from *basis1* to *basis2*. Note that this is a passive rotation of the coordinate system and produces the polarization components with respect to the new basis.

```
TransformationMatrix[basis2_, basis1_] :=
 Table[Dot[basis1⟦i⟧, basis2⟦j⟧], {j, 1, 3}, {i, 1, 3}] // Simplify
```

Reaction basis

The most common basis places the longitudinal direction along the momentum, normal perpendicular to the reaction plane, and transverse in the reaction plane. We denote the *reaction basis* vectors $\{\hat{t}, \hat{n}, \hat{\ell}\}$. Let ϕ represent the dihedral angle between the scattering and reaction planes and θ represent the polar angle between the ejectile momentum and the virtual photon. The basis vectors then become

```
{t̂, n̂, ℓ̂} = Transpose[euler[ϕ, θ, 0]] // Simplify

{{Cos[θ] Cos[ϕ], Cos[θ] Sin[ϕ], -Sin[θ]},
 {-Sin[ϕ], Cos[ϕ], 0}, {Cos[ϕ] Sin[θ], Sin[θ] Sin[ϕ], Cos[θ]}}

euler[ϕ, θ, 0].{0, 0, 1}

{Cos[ϕ] Sin[θ], Sin[θ] Sin[ϕ], Cos[θ]}

euler[ϕ, θ, 0].{1, 0, 0}

{Cos[θ] Cos[ϕ], Cos[θ] Sin[ϕ], -Sin[θ]}

n̂ /. ϕ → 0

{0, 1, 0}

t̂ /. ϕ → 0

{Cos[θ], 0, -Sin[θ]}

t̂ /. ϕ → π/2

{0, Cos[θ], -Sin[θ]}
```

Unfortunately, the reaction basis has the undesirable property that its vertical component reverses sign as $\hat{p}$ moves from one side of $\hat{q}$ to the other.

```
ReactionBasis = {t̂, n̂, ℓ̂};
```

```
ReactionBasis /. ϕ → 0 // Simplify
```

```
{{Cos[θ], 0, -Sin[θ]}, {0, 1, 0}, {Sin[θ], 0, Cos[θ]}}
```

```
ReactionBasis /. ϕ → π // Simplify
```

```
{{-Cos[θ], 0, -Sin[θ]}, {0, -1, 0}, {-Sin[θ], 0, Cos[θ]}}
```

Similarly, $\hat{t}.\hat{x}$ also reverses sign abruptly. These sign reversals are reflected in abrupt changes in the polarization components on either side of $\hat{q}$. Smoother behavior is achieved using the *polarimeter* basis developed below. First, however, we give the matrix which transforms the polarization components from the reaction basis to the photon basis.

```
ReactionToPhoton =
  TransformationMatrix[IdentityMatrix[3], ReactionBasis];
MatrixForm[ReactionToPhoton]
```

$$\begin{pmatrix} \text{Cos}[\theta]\,\text{Cos}[\phi] & -\text{Sin}[\phi] & \text{Cos}[\phi]\,\text{Sin}[\theta] \\ \text{Cos}[\theta]\,\text{Sin}[\phi] & \text{Cos}[\phi] & \text{Sin}[\theta]\,\text{Sin}[\phi] \\ -\text{Sin}[\theta] & 0 & \text{Cos}[\theta] \end{pmatrix}$$

```
ReactionToPhoton.{Pt, Pn, Pℓ} // ColumnForm
```

$$-\text{Sin}[\phi]\,P_n + \text{Cos}[\theta]\,\text{Cos}[\phi]\,P_t + \text{Cos}[\phi]\,\text{Sin}[\theta]\,P_\ell$$
$$\text{Cos}[\phi]\,P_n + \text{Cos}[\theta]\,\text{Sin}[\phi]\,P_t + \text{Sin}[\theta]\,\text{Sin}[\phi]\,P_\ell$$
$$-\text{Sin}[\theta]\,P_t + \text{Cos}[\theta]\,P_\ell$$

Notice that if the acceptance is symmetric wrt the scattering plane, the $\hat{x}$ and $\hat{z}$ components are determined by orthogonal linear combinations of the helicity dependent $\hat{t}$ and $\hat{\ell}$ components while the $\hat{y}$ component is determined by the helicity-independent $\hat{n}$ component of polarization.

Polarimeter basis

■ Derivation based upon vector cross product

In this section we derive an alternative $\{\hat{a},\hat{b},\hat{c}\}$ basis that avoids the inconvenient sign reversals exhibited by the reaction basis as $\hat{p}$ moves from one side of $\hat{q}$ to the other. We call this the polarimeter basis because a nonmagnetic polarimeter with small acceptance would naturally be placed with its central $\hat{c}$ axis along the ejectile momentum and its sideways $\hat{a}$ axis parallel to the scattering plane, which is normally horizontal in the lab. Furthermore, since we would not turn the polarimeter upside down, we would like to keep the vertical component of $\hat{b}$ positive. It is simplect to construct the polarimeter basis using $\hat{c} = \hat{\ell}$, $\hat{a} \propto \hat{y}\times\hat{\ell}$, and $\hat{b} = \hat{c}\times\hat{a}$.

```
ĉ = ℓ̂
```

```
{Cos[ϕ] Sin[θ], Sin[θ] Sin[ϕ], Cos[θ]}
```

```
â = NormalizedCrossProduct[{0, 1, 0}, ℓ̂]
```

$$\left\{\frac{\text{Cos}[\theta]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}},\ 0,\ -\frac{\text{Cos}[\phi]\,\text{Sin}[\theta]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}}\right\}$$

```
b̂ = Cross[ĉ, â] // Simplify
```

$$\left\{-\frac{\text{Cos}[\phi]\,\text{Sin}[\theta]^2\,\text{Sin}[\phi]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}},\ \sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2},\ -\frac{\text{Cos}[\theta]\,\text{Sin}[\theta]\,\text{Sin}[\phi]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}}\right\}$$

```
â.b̂ == b̂.ĉ == â.ĉ == 0 // Simplify
```

```
True
```

In contrast with the reaction basis, the polarimeter basis has the desirable property that its vertical component is always positive and that $\hat{a}.\hat{x}$ does not reverse sign as $\hat{p}$ moves from one side of $\hat{q}$ to the other.

```
PolarimeterBasis = {â, b̂, ĉ};
```

```
PolarimeterBasis /. ϕ → 0 // Simplify
```

```
{{Cos[θ], 0, -Sin[θ]}, {0, 1, 0}, {Sin[θ], 0, Cos[θ]}}
```

```
PolarimeterBasis /. ϕ → π // Simplify
```

```
{{Cos[θ], 0, Sin[θ]}, {0, 1, 0}, {-Sin[θ], 0, Cos[θ]}}
```

■ Derivation using rotation matrix

It is also instructive to express the polarimeter basis in terms of Euler angles. This basis can be obtained from the reaction basis using a final rotation about the $\hat{\ell}$ axis through an angle ψ designed to eliminate $\hat{a}.\hat{y}$. Trial solutions are obtained by solving the following equation.

```
eq = 0 == Part[Transpose[euler[ϕ, θ, ψtemp]], 1, 2]
```

```
0 == Cos[θ] Cos[ψtemp] Sin[ϕ] + Cos[ϕ] Sin[ψtemp]
```

```
sol = Solve[eq, ψtemp]
```

— *Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found.*

$$\left\{\left\{\psi\text{temp}\to-\text{ArcCos}\left[-\frac{\text{Cos}[\phi]}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}\right]\right\},\left\{\psi\text{temp}\to\text{ArcCos}\left[-\frac{\text{Cos}[\phi]}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}\right]\right\},\left\{\psi\text{temp}\to-\text{ArcCos}\left[\frac{\text{Cos}[\phi]}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}\right]\right\},\left\{\psi\text{temp}\to\text{ArcCos}\left[\frac{\text{Cos}[\phi]}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}\right]\right\}\right\}$$

To choose between these solutions we must impose a subsidiary condition. Since we would not turn the polarimeter upside down, we would like to keep the vertical component of the second basis vector positive.

```
by[ψ_] := Part[Transpose[euler[ϕ, θ, ψ]], 2, 2] // Simplify
```

```
bytemp = by[ψtemp] /. sol
```

$$\left\{-\frac{\text{Cos}[\phi]^2}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}+\text{Cos}[\theta]\,\text{Sin}[\phi]\sqrt{1-\frac{\text{Cos}[\phi]^2}{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}},\right.$$
$$-\frac{\text{Cos}[\phi]^2}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}-\text{Cos}[\theta]\,\text{Sin}[\phi]\sqrt{1-\frac{\text{Cos}[\phi]^2}{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}},$$
$$\frac{\text{Cos}[\phi]^2}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}+\text{Cos}[\theta]\,\text{Sin}[\phi]\sqrt{1-\frac{\text{Cos}[\phi]^2}{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}},$$
$$\left.\frac{\text{Cos}[\phi]^2}{\sqrt{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}-\text{Cos}[\theta]\,\text{Sin}[\phi]\sqrt{1-\frac{\text{Cos}[\phi]^2}{\text{Cos}[\phi]^2+\text{Cos}[\theta]^2\,\text{Sin}[\phi]^2}}\right\}$$

Table$\left[\textbf{Sign[bytemp] /. }\left\{\theta\to\frac{\pi}{4}\right\},\left\{\phi,\frac{\pi}{4},7\frac{\pi}{4},\frac{\pi}{4}\right\}\right]$ **// MatrixForm**

$$\begin{pmatrix}-1 & -1 & 1 & 1\\ 1 & -1 & 1 & -1\\ -1 & -1 & 1 & 1\\ -1 & -1 & 1 & 1\\ -1 & -1 & 1 & 1\\ -1 & 1 & -1 & 1\\ -1 & -1 & 1 & 1\end{pmatrix}$$

```
Table[Sign[bytemp] /. {θ → 3π/4}, {ϕ, π/4, 7 π/4, π/4}] // MatrixForm
```

$$\begin{pmatrix} -1 & -1 & 1 & 1 \\ -1 & 1 & -1 & 1 \\ -1 & -1 & 1 & 1 \\ -1 & -1 & 1 & 1 \\ -1 & -1 & 1 & 1 \\ 1 & -1 & 1 & -1 \\ -1 & -1 & 1 & 1 \end{pmatrix}$$

Unfortunately, it is not possible to obtain this behavior from a single solution; the form of these solutions also presents problems when **PowerExpand** is used to simplify subsequent results. Recognizing that for most orientations either solution 3 or solution 4 yields the desired behavior, we attempt to find a simpler solution using **PowerExpand**

```
ψ3 = ψtemp /. sol⟦3⟧;
{Cos[ψ3], Sin[ψ3], Tan[ψ3]} // Simplify // PowerExpand
```

$$\left\{\frac{\mathrm{Cos}[\phi]}{\sqrt{\mathrm{Cos}[\phi]^2+\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\phi]^2}},\ -\frac{\mathrm{Cos}[\theta]\,\mathrm{Sin}[\phi]}{\sqrt{\mathrm{Cos}[\phi]^2+\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\phi]^2}},\ -\mathrm{Cos}[\theta]\,\mathrm{Tan}[\phi]\right\}$$

```
ψ4 = ψtemp /. sol⟦4⟧;
{Cos[ψ4], Sin[ψ4], Tan[ψ4]} // Simplify // PowerExpand
```

$$\left\{\frac{\mathrm{Cos}[\phi]}{\sqrt{\mathrm{Cos}[\phi]^2+\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\phi]^2}},\ \frac{\mathrm{Cos}[\theta]\,\mathrm{Sin}[\phi]}{\sqrt{\mathrm{Cos}[\phi]^2+\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\phi]^2}},\ \mathrm{Cos}[\theta]\,\mathrm{Tan}[\phi]\right\}$$

Actually, it is more efficient to identify {Cos[ψ], Sin[ψ]} with the projections of $\hat{a}$ upon $\hat{t}$ and $\hat{n}$, whereby

```
{â.t̂, â.n̂} // Simplify
```

$$\left\{\frac{\mathrm{Cos}[\phi]}{\sqrt{\mathrm{Cos}[\theta]^2+\mathrm{Cos}[\phi]^2\,\mathrm{Sin}[\theta]^2}},\ -\frac{\mathrm{Cos}[\theta]\,\mathrm{Sin}[\phi]}{\sqrt{\mathrm{Cos}[\theta]^2+\mathrm{Cos}[\phi]^2\,\mathrm{Sin}[\theta]^2}}\right\}$$

Thus, we demonstrate that the following representation does ensure a positive vertical component.

```
ψsol = ArcTan[Cos[ϕ], -Cos[θ] Sin[ϕ]];

bytemp2 = by[ψsol];
```

```
Table[Sign[bytemp2] /. {θ → π/4}, {ϕ, π/4, 7 π/4, π/4}] // MatrixForm
```

$$\begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

```
Table[Sign[bytemp2] /. {θ → 3 π/4}, {ϕ, π/4, 7 π/4, π/4}] // MatrixForm
```

$$\begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

Therefore, although it is difficult to obtain simple symbolic results, we have found a symbolic expression that appears to give the desired rotation for any choice of θ and ϕ.

```
Plot[Evaluate[Table[ψsol, {θ, 0, π/2, π/16}]], {ϕ, 0, 2 π}];
```



```
Plot3D[ψsol, {θ, 0, π}, {ϕ, 0, 2 π},
  PlotPoints → 25, ViewPoint -> {-2.224, -2.259, 1.184}];
```



We can transform from the reaction basis to the polarimeter basis by rotation about $\hat{\ell}$ through angle ψ.

```
ReactionToPolarimeter =
  TransformationMatrix[PolarimeterBasis, ReactionBasis];
MatrixForm[ReactionToPolarimeter]
```

$$\begin{pmatrix} \frac{\text{Cos}[\phi]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}} & -\frac{\text{Cos}[\theta]\,\text{Sin}[\phi]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}} & 0 \\ \frac{\text{Cos}[\theta]\,\text{Sin}[\phi]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}} & \frac{\text{Cos}[\phi]}{\sqrt{\text{Cos}[\theta]^2+\text{Cos}[\phi]^2\,\text{Sin}[\theta]^2}} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Although I seem to have lost my earlier notes on this derivation, the following verifies that this result agrees with the polarimeter basis coded in *epiprod*.

```
Cos[ϕ]^2 + Cos[θ]^2 Sin[ϕ]^2 == 1 - Sin[θ]^2 Sin[ϕ]^2 // Simplify
```

```
True
```

Spectrometer basis

The spectrometer basis $\{\hat{d}, \hat{e}, \hat{f}\}$ is similar to the photon basis in that $\hat{e} = \hat{y}$ is vertical, but now we set $\hat{f}$ at some angle θ_s wrt to the nominal $\hat{q}$.

```
{d̂, ê, f̂} = Transpose[euler[0, θs, 0]] // Simplify;
SpectrometerBasis = {d̂, ê, f̂}
```

```
{{Cos[θs], 0, -Sin[θs]}, {0, 1, 0}, {Sin[θs], 0, Cos[θs]}}
```

```
ReactionToSpectrometer =
 TransformationMatrix[SpectrometerBasis, ReactionBasis]
```

```
{{Cos[θ] Cos[θs] Cos[ϕ] + Sin[θ] Sin[θs],
  -Cos[θs] Sin[ϕ], Cos[θs] Cos[ϕ] Sin[θ] - Cos[θ] Sin[θs]},
 {Cos[θ] Sin[ϕ], Cos[ϕ], Sin[θ] Sin[ϕ]},
 {-Cos[θs] Sin[θ] + Cos[θ] Cos[ϕ] Sin[θs],
  -Sin[θs] Sin[ϕ], Cos[θ] Cos[θs] + Cos[ϕ] Sin[θ] Sin[θs]}}
```

```
ReactionToSpectrometer.{Pt, Pn, Pℓ} // Simplify
```

```
{-Cos[θs] Sin[ϕ] Pn + (Cos[θ] Cos[θs] Cos[ϕ] + Sin[θ] Sin[θs]) Pt +
   (Cos[θs] Cos[ϕ] Sin[θ] - Cos[θ] Sin[θs]) Pℓ,
 Cos[ϕ] Pn + Sin[ϕ] (Cos[θ] Pt + Sin[θ] Pℓ),
 -Sin[θs] Sin[ϕ] Pn + (-Cos[θs] Sin[θ] + Cos[θ] Cos[ϕ] Sin[θs]) Pt +
   (Cos[θ] Cos[θs] + Cos[ϕ] Sin[θ] Sin[θs]) Pℓ}
```

```
SpectrometerToReaction =
 TransformationMatrix[ReactionBasis, SpectrometerBasis]
```

```
{{Cos[θ] Cos[θs] Cos[ϕ] + Sin[θ] Sin[θs],
  Cos[θ] Sin[ϕ], -Cos[θs] Sin[θ] + Cos[θ] Cos[ϕ] Sin[θs]},
 {-Cos[θs] Sin[ϕ], Cos[ϕ], -Sin[θs] Sin[ϕ]},
 {Cos[θs] Cos[ϕ] Sin[θ] - Cos[θ] Sin[θs],
  Sin[θ] Sin[ϕ], Cos[θ] Cos[θs] + Cos[ϕ] Sin[θ] Sin[θs]}}
```

```
SpectrometerToReaction.{Pd, Pe, Pf} // Simplify
```

```
{(Cos[θ] Cos[θs] Cos[ϕ] + Sin[θ] Sin[θs]) Pd +
  Cos[θ] Sin[ϕ] Pe + (-Cos[θs] Sin[θ] + Cos[θ] Cos[ϕ] Sin[θs]) Pf,
 -Cos[θs] Sin[ϕ] Pd + Cos[ϕ] Pe - Sin[θs] Sin[ϕ] Pf,
 (Cos[θs] Cos[ϕ] Sin[θ] - Cos[θ] Sin[θs]) Pd +
  Sin[θ] Sin[ϕ] Pe + (Cos[θ] Cos[θs] + Cos[ϕ] Sin[θ] Sin[θs]) Pf}
```